

Excel

Fonctions intégrées



Généralités

Fonctions
mathématiques

Fonctions
statistiques

Fonctions logiques

Fonctions
d'information


Fonctions
de recherche
et référence

Fonctions
de date et heure

Fonctions
de texte



- Excel est fourni avec un catalogue comportant plusieurs dizaines de fonctions de ce fait appelées « fonctions intégrées ».
- Le nom d'une fonction ne comporte jamais d'espace. Certains noms de fonctions comportent un point (par exemple NB.SI).
- Le nom d'une fonction est toujours suivi de parenthèses (ouvrante et fermante).
- Les parenthèses peuvent accueillir zéro, un ou plusieurs éléments appelés *arguments de la fonction* : ce sont les paramètres requis pour utiliser la fonction.
- Le point-virgule est utilisé pour séparer les arguments d'une fonction
- Le nombre et l'ordre des arguments peut être variable ou imposé. On parle ainsi d'arguments *obligatoires* ou *facultatifs*. Lorsqu'un ordre est requis, les arguments sont généralement nommés.



Le groupe Bibliothèque de fonctions dans l'onglet Formules : les catégories utilisées dans ce document (cf. le sommaire situé page précédente) correspondent aux menus ci-dessus.

=AUJOURDHUI()

La fonction AUJOURDHUI() n'a pas d'argument.

=SOMME(B2:B10)

L'argument B2:B10 figure entre parenthèses.

=MAX(B2:B8;D2:D8)

Les arguments B2:B8 et D2:D8 sont séparés par un point-virgule.

=ARRONDI(B2;2)
ARRONDI(nombre; no_chiffres)

La fonction ARRONDI() possède 2 arguments nommés obligatoires : *nombre* et *no_chiffres*.

=EQUATION.RANG(B2;B2:B10)
EQUATION.RANG(nombre; référence; [ordre])

La fonction EQUATION.RANG() possède 3 arguments nommés : les 2 premiers sont obligatoires. Le caractère facultatif du 3^e argument est signalé par les crochets qui l'encadrent.

SOMME

- Cette fonction effectue la somme des valeurs numériques désignées.
- Cette fonction ignore les données non numériques (texte, valeurs logiques) et les cellules vides.
- Elle renvoie une valeur d'erreur dès que l'un de ses arguments contient une valeur d'erreur. On utilise dans ce cas la fonction SOMME.SI*.
- Cette fonction possède 1 à 255 arguments.

PRODUIT

- Cette fonction multiplie les valeurs désignées comme arguments.
- Si l'une des valeurs désignées est 0, le résultat est 0 ; si l'une des valeurs désignées est une valeur d'erreur, le résultat est une valeur d'erreur.
- Cette fonction possède 1 à 255 arguments

SOMME(nombre1; [nombre2]; ...)

=SOMME(A2:A15)	Somme des valeurs numériques contenues dans la plage.
=SOMME(A2:A15;C2:C15)	Somme des valeurs numériques contenues dans les 2 plages non adjacentes.

PRODUIT(nombre1;[nombre2]; ...)

=PRODUIT(A2:C2)	Équivaut à $A2*B2*C2$.
=PRODUIT(A2;C2;E2;G2)	Équivaut à $A2*C2*E2*G2$.

ARRONDI

- Cette fonction arrondit un nombre au plus proche et au chiffre spécifié.
- L'argument *no_chiffres* arrondit la partie décimale du nombre lorsqu'il est positif ; il arrondit à l'unité s'il vaut zéro et arrondit la partie entière du nombre s'il est négatif (-1, aux dizaines, -2 aux centaines, -3 aux milliers, etc.)

ENT

- Cette fonction renvoie la partie entière d'un nombre.

ABS

- Cette fonction renvoie la valeur absolue d'un nombre, c'est-à-dire un nombre sans son signe.

ARRONDI(nombre; no_chiffres)

=ARRONDI(8,753;2) =ARRONDI(4,234;2)	Arrondi au centième soit 8,75 et 4,23
=ARRONDI(8,753;1) =ARRONDI(4,234;1)	Arrondi au dixième soit 8,8 et 6,2
=ARRONDI(8,753;0) =ARRONDI(4,234;0)	Arrondi à l'unité soit 9 et 4
=ARRONDI(8,753;-1) =ARRONDI(4,234;-1)	Arrondi à la dizaine soit 10 et 0

ENT(nombre)

=ENT(7,58)	Résultat : 7.
=ENT("01/07/2017 14:30:00")	Résultat : 42917, c'est-à-dire le numéro de série correspondant à la date du 1 ^{er} juillet 2017, sans les heures, minutes, secondes.

ABS(nombre)

=ABS(-15)	Résultat : 15.
-----------	----------------

SOMME.SI

- Cette fonction additionne, dans une plage, le nombre de cellules répondant au critère.
- Les 2 premiers arguments de cette fonction sont obligatoires, le 3^e est facultatif.
- Le critère est exprimé sous forme de texte, de nombre ou d'expression pouvant comporter des références.
- Le critère porte toujours sur les valeurs indiquées dans l'argument *plage*.
- L'argument *somme_plage* est requis lorsqu'on souhaite additionner des valeurs autres que celles désignées par l'argument *plage* : si la 1^{re} valeur de l'argument *plage* répond au critère, la 1^{re} valeur de l'argument *somme_plage* est incluse dans le résultat et ainsi de suite pour les autres valeurs.
- Cette fonction admet un seul critère.
- Pour effectuer une opération similaire en combinant plusieurs critères, il convient d'utiliser la fonction SOMME.SI.ENS*.

SOMME.SI(plage; critère; [somme_plage])

=SOMME.SI(A2:A15;20)

Additionne dans la plage les valeurs égales à 20. On notera que l'opérateur = est omis.

=SOMME.SI(A2:A15;"=20")

Additionne dans la plage les valeurs égales à 20. On notera que l'opérateur = est ici exprimé ce qui fait du critère une expression assimilée à une constante textuelle (d'où les guillemets).

=SOMME.SI(A2:A15;">20")

Additionne dans la plage les valeurs égales à 20. On notera que l'opérateur > fait du critère une expression assimilée à une constante textuelle (d'où les guillemets).

=SOMME.SI(A2:A15;"01/04/2017";B2:B15)

Additionne les cellules de la plage B2:B15 (argument *somme_plage*) à condition que la date correspondante de la plage A2:A15 soit le 01/04/2017. La constante de type date figure entre guillemets.

=SOMME.SI(A2:A15;">"&C12)

Additionne les cellules dont la valeur est strictement supérieure à la valeur indiquée dans la cellule C12. Le critère concatène* une constante et une référence grâce à l'opérateur &.

L'opérateur est une constante de type texte et, à ce titre, placé entre guillemets. La référence C12 n'est par nature pas entre guillemets.

SOMME.SI.ENS

- Cette fonction additionne des valeurs relativement à une ou plusieurs conditions.
- L'argument *plage_somme* désigne les valeurs à additionner.
- L'argument *plage_critères1* désigne les valeurs sur lesquelles porte le 1^{er} critère. Cette plage est généralement différente de celle désignée dans l'argument *plage_somme*.
- L'argument *critères1* désigne le critère appliqué aux valeurs de l'argument *plage_critères1*.
- Si plusieurs critères sont définis pour une ou plusieurs plages, la fonction évalue si la 1^{re} cellule de chaque plage de critères répond à son critère et, dans l'affirmative, elle retient la 1^{re} valeur de l'argument *plage_somme* pour le résultat. Elle procède de même pour la 2^e cellule de chaque plage et ainsi de suite.
- Pour en savoir davantage sur les expressions de critères, consulter la description de la fonction SOMME.SI*.
- Elle comporte entre 1 et 127 couples d'arguments *plage_critères/critères*.

SOMME.SI.ENS(plage_somme; plage_critères1; critères1; [plage_critères2; critères2]; ...)

=SOMME.SI.ENS(C2:C12;A2:A12;"Semaine 14";B2:B12;"Antonio")

Semaine	Vendeur	Montant
Semaine 14	Antonio	26355,13
Semaine 14	Lætitia	38934
Semaine 14	Antonio	27520,31
Semaine 14	Malik	21630
Semaine 14	Étienne	23419,63
Semaine 15	Claire	31769,06
Semaine 15	Antonio	41444,63
Semaine 15	Malik	38934
Semaine 15	Claire	36153
Semaine 15	Lætitia	27037,5
Semaine 15	Malik	40195,75

Cette formule additionne les montants se trouvant sur des lignes pour lesquelles, simultanément, le vendeur est *Antonio* et la semaine est *Semaine 14*.

C2:C12 désigne la plage des montants.

A2:A12 désigne la plage des Semaines sur laquelle porte le critère "Semaine 14".

B2:B12 désigne la plage des prénoms sur laquelle porte le critère "Antonio".

Cette formule additionne le 1^{er} et le 3^e montant et son résultat est $26355,13 + 27520,31 = 53875,44$.

SOMMEPROD

- Cette fonction additionne les produits de valeurs issues de plusieurs matrices ou plages de cellules.
- Chaque valeur de l'argument *matrice1* est multiplié par son homologue (valeur occupant la même position) de la *matrice2*, puis de la *matrice3*, etc.
La fonction renvoie ensuite la somme de ces différents produits.
- Les différentes matrices doivent être de même étendue et sont généralement désignées par des plages de cellules.
- Cette fonction comporte 2 à 255 matrices.

SOMMEPROD(matrice1; [matrice2]; ...)

=SOMMEPROD({5.4};{6.3})

Cet exemple montre 2 constantes matricielles (entre accolades) comme arguments : {5.4} désigne une matrice comportant 2 valeurs présentées sur une même ligne ; *idem* pour la constante matricielle {6.3}.

Le 1^{er} nombre de la 1^{re} matrice est multiplié par le 1^{er} nombre de la 2^e matrice ;
le 2^e nombre de la 1^{re} matrice est multiplié par le 2^e nombre de la 2^e matrice ;
enfin, ces 2 produits sont additionnés, soit :
 $5 \times 6 + 4 \times 3 = 30 + 12 = 42$.

=SOMMEPROD(A1:B1;A2:B2)

5	4
6	3

A1 est multiplié par A2 ; B1 est multiplié par B2 puis ces 2 produits sont additionnés, soit :
 $A1 \times A2 + B1 \times B2 = 5 \times 6 + 4 \times 3 = 30 + 12 = 42$

SOUS.TOTAL

- Cette fonction effectue des calculs statistiques sur des plages de données en tenant compte des filtres appliqués et, le cas échéant, des lignes masquées manuellement (par exemple par l'onglet **Accueil**, groupe **Cellules**, menu **Format**, sous-menu **Masquer & Afficher**, commande **Masquer**).
- L'argument *no_fonction* indique la fonction statistique à utiliser (cf. le tableau ci-contre) :
 - les fonctions désignées par les numéros 1 à 11 ne sont pas affectées par les lignes masquées manuellement ;
 - les fonctions désignées par les numéros de 101 à 111 le sont.
- Toutes les fonctions (de 1 à 11, d'une part, et de 101 à 111, d'autre part) sont affectées par les filtres.
- La fonction SOUS.TOTAL ignore les valeurs issues d'autres fonctions SOUS.TOTAL éventuellement présentes dans les plages désignées pour éviter les doubles comptes.

SOUS.TOTAL(no_fonction; réf1; [réf2]; ...)

=SOUS.TOTAL(9;C2:C76)

Effectue la somme (numéro de fonction 9) des valeurs contenues dans la plage sans être affectée par le fait de masquer manuellement certaines lignes de la feuille de calcul.

=SOUS.TOTAL(103;C2:C76)

Évalue le nombre de cellules non vides et visibles dans la plage, qu'il s'agisse de lignes masquées manuellement ou grâce à un filtre.

<i>Incluent les valeurs masquées manuellement</i>	<i>Ignorent les valeurs masquées manuellement</i>	<i>Fonction</i>
1	101	MOYENNE
2	102	NB
3	103	NBVAL
4	104	MAX
5	105	MIN
6	106	PRODUIT
7	107	ECARTYPE
8	108	ECARTYPEP
9	109	SOMME
10	110	VAR
11	111	VAR.P

MOYENNE

- Cette fonction évalue la moyenne arithmétique des valeurs numériques désignées comme arguments.
- Cette fonction ignore les données non numériques (texte, valeurs logiques, valeurs d'erreur) et les cellules vides.
- Cette fonction possède 1 à 255 arguments.

MAX

- Cette fonction renvoie la plus grande des valeurs désignées comme arguments.
- Si la sélection ne contient que du texte ou des cellules vides, elle renvoie 0.
- Cette fonction possède 1 à 255 arguments.

MIN

- Cette fonction renvoie la plus petite des valeurs désignées comme arguments.
- Si la sélection ne contient que du texte ou des cellules vides, elle renvoie 0.
- Cette fonction possède 1 à 255 arguments.

MOYENNE(nombre1; [nombre2]; ...)

=MOYENNE(A2:A15)

Moyenne arithmétique des valeurs numériques contenues dans la plage (somme des valeurs numériques divisée par le nombre de valeurs numériques).

=MOYENNE(A2:A15;C2:C15)

Moyenne arithmétique des valeurs numériques contenues dans les 2 plages non adjacentes.

MAX(nombre1;[nombre2]; ...)

=MAX(A2:A15)

Plus grande valeur dans la plage.

=MAX(A2:A15;C2:C15)

Plus grande valeur dans 2 plages non adjacentes.

MIN(nombre1;[nombre2]; ...)

=MIN(A2:A15)

Plus petite valeur dans la plage.

=MIN(A2:A15;C2:C15)

Plus petite valeur dans 2 plages non adjacentes.

NB

- Cette fonction compte, dans une ou plusieurs plages, le nombre de cellules contenant des valeurs numériques.
- Cette fonction ignore les données non numériques (texte, valeurs logiques, valeurs d'erreur) et les cellules vides.
- Cette fonction possède 1 à 255 arguments.

NBVAL

- Cette fonction compte, dans une plage ou plusieurs plages, le nombre de cellules non vides.
- Cette fonction possède 1 à 255 arguments.

NB.VIDE

- Cette fonction compte, dans une seule plage de cellules, le nombre de cellules vides.
- Cette fonction possède un seul argument obligatoire.

NB(valeur1; [valeur2]; ...)

=NB(A2:A15)

Nombre de cellules dans la plage contenant des valeurs numériques.

=NB(A2:A15;C2:C15)

Nombre de cellules, dans 2 plages non adjacentes, contenant des valeurs numériques

NBVAL(valeur1; [valeur2]; ...)

=NBVAL(A2:A15)

Nombre de cellules non vides dans la plage.

=NBVAL(A2:A15;C2:C15)

Nombre de cellules non vides dans 2 plages non adjacentes.

NB.VIDE(plage)

=NB.VIDE(A2:A15)

Nombre de cellules vides dans la plage A2:A15

NB.SI

- Cette fonction compte, dans une plage, le nombre de cellules répondant au critère.
- Les arguments de cette fonction sont obligatoires.
- Le critère est exprimé sous forme de texte, de nombre ou d'expression (une combinaison d'éléments) pouvant comporter des références.
- Cette fonction possède un seul critère.
- Pour effectuer une opération similaire en combinant plusieurs critères, il convient d'utiliser la fonction NB.SI.ENS*.

NB.SI(plage; critère)

=NB.SI(A2:A15;20)	Nombre de cellules dans la plage dont la valeur est égale à 20. On notera que l'opérateur de comparaison « = » est omis.
=NB.SI(A2:A15;"=20")	Nombre de cellules dans la plage dont la valeur est égale à 20. On notera que l'opérateur « = » est ici exprimé ce qui fait du critère une expression assimilée à une constante textuelle (d'où les guillemets).
=NB.SI(A2:A15;">20")	Nombre de cellules dans la plage dont la valeur est supérieure à 20. On notera que l'opérateur > fait du critère une expression assimilée à une constante textuelle (d'où les guillemets).
=NB.SI(A2:A15;"01/04/2017")	Nombre de cellules dans la plage dont la valeur correspond à la date saisie en critère. La constante de type date est placée entre guillemets.
=NB.SI(A2:A15;C12)	Nombre de cellules dans la plage dont la valeur correspond à la valeur indiquée dans la cellule C12.
=NB.SI(A2:A15;">"&C12)	Nombre de cellules dans la plage dont la valeur est strictement supérieure à la valeur indiquée dans la cellule C12. Le critère concatène* une constante et une référence grâce à l'opérateur &. L'opérateur est une constante de type texte et, à ce titre, placé entre guillemets. La référence C12 n'est par nature pas entre guillemets.

NB.SI.ENS

- Cette fonction compte, dans une ou plusieurs plages, combien de fois tous les critères sont remplis.
- Si plusieurs critères sont appliqués à une même plage, la fonction évalue combien de cellules de cette plage répondent simultanément à tous les critères.
- Si plusieurs critères sont appliqués à plusieurs plages, la fonction évalue si la 1^{re} valeur de chaque plage répond à son critère et compte 1 seulement si les critères sont respectés par toutes les 1^{res} valeurs simultanément et ainsi de suite pour la 2^e valeur de chaque plage, etc.
- Elle comporte entre 1 et 127 couples d'arguments plage_critères/critères.
- Pour en savoir davantage sur la formulation des critères, consulter la description de la fonction NB.SI*.

NB.SI.ENS(plage_critères1; critères1; [plage_critères2; critères2]; ...)

=NB.SI.ENS(A1:A5;">=10";A1:A5;"<=20")

10
16
5
12
25

Cette formule évalue combien de valeurs situées dans la plage A1:A5 sont comprises entre 10 et 20, c'est-à-dire simultanément :

- supérieures ou égales à 10 ;
- inférieures ou égales à 20.

=NB.SI.ENS(A1:A5;">10";B1:B5;">20")

10	17
16	23
5	27
12	31
25	12

Cette formule compte combien de lignes contiennent simultanément :

- un nombre supérieur à 10 dans la colonne de gauche ;
- un nombre supérieur à 20 dans la colonne de droite.

Le résultat est 2 car seules les 2^e et avant-dernière lignes répondent à ces critères.

EQUATION.RANG

- Cette fonction évalue la position d'une valeur relativement à un groupe de valeurs.
- L'argument *nombre* est la valeur dont on veut connaître le rang.
- L'argument *référence* est l'ensemble des valeurs prises en compte pour le classement (cet argument inclut la valeur indiquée dans l'argument *nombre*).
- L'argument facultatif *ordre* indique s'il s'agit d'un classement par ordre décroissant (on peut alors ne pas le compléter ou indiquer 0) ou par ordre croissant (on le complète dans ce cas par un 1 ou toute valeur différente de 0).

EQUATION.RANG(nombre; référence; [ordre])

=EQUATION.RANG(A2;\$A\$2:\$A\$5)
=EQUATION.RANG(A2;\$A\$2:\$A\$5;0)

Score	Rang
123	2
97	4
152	1
121	3

Cette formule évalue la position (rang) de la valeur « 123 » par rapport à l'ensemble des valeurs présentées dans la colonne de gauche du tableau ci-contre.

L'argument *ordre* peut être omis ou complété avec la valeur 0 car l'ordre de classement est décroissant (le meilleur score et le score le plus élevé).

=EQUATION.RANG(A2;\$A\$2:\$A\$5;1)

Temps	Rang
01:08	4
00:57	2
00:53	1
01:02	3

Cette formule évalue la position (rang) de la valeur « 01:08 » par rapport à l'ensemble des valeurs présentées dans la colonne de gauche du tableau ci-contre.

L'argument *ordre* a été complété par 1 (ou toute valeur différente de 0) car l'ordre de classement est croissant (le meilleur temps est le temps le plus court).

GRANDE.VALEUR

- Cette fonction renvoie, parmi un groupe de valeurs, celle qui occupe un rang déterminé en partant de la plus grande valeur.
- L'argument *matrice* désigne les valeurs à étudier (référence ou constante matricielle).
- L'argument *k* indique le rang occupé par la valeur à renvoyer comme résultat, à partir de la plus grande valeur.
- Si l'argument *k* vaut 1, cette fonction équivaut à la fonction MAX*.

PETITE.VALEUR

- Cette fonction renvoie, parmi un groupe de valeurs, celle qui occupe un rang déterminé en partant de la plus petite valeur.
- L'argument *matrice* désigne les valeurs à étudier (référence ou constante matricielle).
- L'argument *k* indique le rang occupé par la valeur à renvoyer comme résultat, à partir de la plus petite valeur.
- Si l'argument *k* vaut 1, cette fonction équivaut à la fonction MIN*.

GRANDE.VALEUR(matrice; k)

=GRANDE.VALEUR(A2:A5;3) Cette formule renvoie 121 car il s'agit du 3^e score le plus élevé.

Score
123
97
152
121

PETITE.VALEUR(matrice;k)

=PETITE.VALEUR(A2:A5;3) Cette formule renvoie 14 car il s'agit de la 3^e note la plus faible.

Note
9
14
17
12

SI

- Cette fonction renvoie un résultat si une condition exprimée dans l'argument *test_logique* est respectée et un autre résultat dans le cas contraire.
- L'argument *test_logique* doit obligatoirement renvoyer une valeur logique VRAI ou FAUX. De ce fait, il est généralement exprimé sous forme de comparaison.
- L'argument *test_logique* peut combiner plusieurs valeurs logiques, en particulier grâce aux fonction ET* et OU*.
- Les arguments [*valeur_si_vrai*] et [*valeur_si_faux*] sont tous deux notés comme facultatifs – ils figurent entre crochets – mais ne peuvent pas être omis simultanément.

SIERREUR

- Cette fonction affiche une valeur personnalisée en remplacement d'une valeur d'erreur telle que #VALEUR!, #DIV/0!, #REF!, #N/A, etc.
- Cette fonction n'est pas prise en charge par les versions antérieures à *Excel 2007*.


SI(test_logique; [valeur_si_vrai]; [valeur_si_faux])

=SI(A2>500;"Offert";"Facturé")

Si la cellule A2 contient une valeur supérieure à 500 alors le résultat est « Offert » sinon le résultat est « Facturé ».

=SI(A2>500;"Offert";SI(A2>200;"Remise";"Facturé"))

Si la cellule A2 contient une valeur supérieure à 500 alors le résultat est « Offert » sinon, la 2^e fonction SI teste si la valeur de A2 est supérieure à 200 et, dans ce cas, le résultat est « Remise » ; sinon (si la valeur de A2 ne répond à aucune des 2 conditions), le résultat est « Facturé ».



SIERREUR(valeur; valeur_si_erreur)

=SIERREUR(A2/B2;"Données manquantes")

Le texte « Données manquantes » est affiché si la division de A2 par B2 provoque une erreur (si B2 est vide ou n'est pas un nombre, par exemple).

Dans le cas contraire, le résultat de la division s'affiche.

SI.NON.DISP

- Cette fonction affiche une valeur personnalisée en remplacement de la valeur d'erreur #N/A, c'est-à-dire d'une valeur non disponible.
- Cette fonction se distingue de la fonction SIERREUR* dans la mesure où elle opère exclusivement avec la valeur d'erreur #N/A.
- Cette fonction est disponible depuis la version *Excel 2013*.

NON

- Cette fonction renvoie une valeur logique VRAI ou FAUX.
- Elle inverse les valeurs logiques : elle renvoie VRAI si l'argument est FAUX et renvoie FAUX si l'argument est VRAI.
- Cette fonction utilise une valeur logique comme argument : ce dernier est donc généralement une comparaison, le résultat d'une fonction logique* telle que ET ou encore OU, le résultat d'une fonction d'information* telle que ESTNA, ESTNUM, ESTVIDE, etc.

SI.NON.DISP(valeur; valeur_si_na)

=SI.NON.DISP(RECHERCHEV(A2;D2:E6;2);"")

Si la fonction RECHERCHEV n'a pas de résultat disponible et renvoie la valeur d'erreur #N/A, la fonction SI.NON.DISP a pour résultat une chaîne de texte vide signalée par "".

NON(valeur_logique)

=SI(NON(ESTVIDE(B2));A2/B2;"")

Si la cellule B2 n'est pas vide, effectuer le calcul sinon renvoyer une chaîne de texte vide (signalée par "").

ET

- La fonction ET combine des valeurs logiques et renvoie elle-même une valeur logique.
- Son résultat est VRAI lorsque tous ses arguments sont VRAI.
- La fonction ET possède 1 à 255 arguments.

OU

- La fonction OU combine des valeurs logiques et renvoie elle-même une valeur logique.
- Son résultat est VRAI dès que l'un de ses arguments est VRAI.
- La fonction OU possède 1 à 255 arguments.

ET(valeur_logique1; [valeur_logique2]; ...)

=ET(VRAI;VRAI;VRAI)	Cette formule renvoie VRAI car tous les arguments sont VRAI.
=ET(VRAI;VRAI;FAUX)	Cette formule renvoie FAUX car l'un des arguments est FAUX.
=ET(A2>10;A2<20)	Les arguments de la fonction ET sont des valeurs logiques généralement issues de comparaisons.
=SI(ET(A2>10;A2<20);"Oui";"Non")	La fonction ET peut être utilisée dans l'argument <i>test_logique</i> de la fonction SI.

OU(valeur_logique1; [valeur_logique2]; ...)

=OU(FAUX;FAUX;VRAI)	La formule renvoie VRAI car l'un des arguments est VRAI.
=OU(FAUX;FAUX;FAUX)	La formule renvoie FAUX car tous les arguments sont FAUX.
=OU(A2>10;A2<20)	Les arguments de la fonction OU sont des valeurs logiques généralement issues de comparaisons.
=SI(OU(A2<10;A2>20);"Non";"Oui")	La fonction OU peut être utilisée dans l'argument <i>test_logique</i> de la fonction SI.

ESTVIDE

- La fonction ESTVIDE renvoie une valeur logique VRAI ou FAUX.
- Son résultat est VRAI si la cellule désignée par son argument est vide.
- Son résultat est FAUX que la cellule désignée par son argument contienne une constante ou une formule, même si le résultat de cette formule est une chaîne de caractères vide.

ESTNA

- La fonction ESTNA renvoie une valeur logique VRAI ou FAUX.
- Son résultat est VRAI si son argument est la valeur d'erreur #N/A.

ESTNUM

- La fonction ESTNUM renvoie une valeur logique VRAI ou FAUX.
- Son résultat est VRAI si son argument est un nombre ou une valeur assimilable à un nombre, telle qu'une date, une heure et FAUX dans le cas contraire.

ESTVIDE(valeur)

=ESTVIDE(A2)	L'argument peut être une référence de cellule.
=ESTVIDE(Taux)	L'argument peut être le nom d'une cellule nommée.
=SI(ESTVIDE(B2);"";A2/B2)	La fonction ESTVIDE renvoie une valeur logique : elle peut être utilisée comme argument <i>test_logique</i> de la fonction SI. La formule ci-contre teste si la cellule B2 est vide. Dans l'affirmative, le résultat de la formule est une chaîne de caractères vide signalée par "". Dans le cas contraire, la fonction SI effectue une division. On évite ainsi l'affichage de la valeur d'erreur #DIV/0! lorsque la cellule B2 est vide.

ESTNA(valeur)

=SI(ESTNA(RECHERCHEV(A2;D2:E6;2));""; RECHERCHEV(A2;D2:E6;2))	La formule ci-dessus affiche une chaîne vide (signalée par "") lorsque la fonction RECHERCHEV n'a pas de résultat (et retourne donc la valeur d'erreur #N/A) et exécute la RECHERCHEV dans le cas contraire. La fonction SI.NON.DISP* simplifie la syntaxe pour ce cas.
---	---

ESTNUM(valeur)

=ESTNUM(A2)	L'argument peut être une référence de cellule.
=ESTNUM(Taux)	L'argument peut être une cellule nommée
=ESTNUM(A2/B2)	L'argument peut être une opération, une fonction intégrée.

RECHERCHEV

- Dans « RECHERCHEV », la lettre V signifie « verticale ».
- Cette fonction utilise généralement 2 tableaux : un 1^{er} tableau à compléter par le résultat de la formule ; un 2nd tableau contenant la valeur à renvoyer comme résultat. Ce 2nd tableau est appelé « tableau de correspondance ».
- La fonction RECHERCHEV compte 4 arguments ; seul le 4^e est facultatif.
- L'argument *valeur_cherchée* désigne la valeur à retrouver dans la 1^{re} colonne du tableau de correspondance.
- L'argument *table_matrice* désigne le tableau de correspondance.
- L'argument *no_index_col* utilise toujours un nombre entier positif désignant le rang de la colonne du tableau de correspondance contenant la valeur à renvoyer.
- L'argument *valeur_proche* n'est pas complété (ou complété par VRAI, de manière facultative) lorsque la 1^{re} colonne du tableau de correspondance contient des valeurs représentant des seuils ou paliers.
- La 1^{re} colonne du tableau de correspondance ne comporte normalement pas de doublons.

RECHERCHEV(valeur_cherchée; table_matrice; no_index_col; [valeur_proche])

=RECHERCHEV(A2;D2:E6;2)

Distance	Port
0	0,00 €
10	5,00 €
20	10,00 €
50	15,00 €
100	20,00 €

La 1^{re} colonne du tableau de correspondance placé en D2:E6 contient des seuils kilométriques permettant de calculer des frais de port.

Ces paliers de facturation doivent obligatoirement être triés dans l'ordre croissant.

Si, par exemple, la valeur indiquée en A2 est 86, étant donné que cette valeur dépasse le seuil 50 mais n'atteint pas le palier 100, alors la formule renvoie une valeur située sur la même ligne que le palier 50. En l'occurrence, puisque l'argument *no_index_col* est 2, le résultat est « 15 » car cette valeur se situe dans la 2^e cellule de la ligne où se situe le palier 50.

Étant donné que la 1^{re} colonne du tableau de correspondance contient des valeurs représentant des seuils, la valeur indiquée comme *valeur_cherchée* n'existe pas nécessairement dans cette colonne et l'argument *valeur_proche* est omis.

=RECHERCHEV(A2;D2:E8;2;FAUX)

Jour	Heure entrée
Lundi	08:00
Mardi	09:00
Mercredi	08:00
Jeudi	07:30
Vendredi	08:00
Samedi	
Dimanche	

La 1^{re} colonne du tableau de correspondance placé en D2:E8 contient les jours de la semaine : cette liste n'est pas triée dans l'ordre alphabétique mais chronologique : de ce fait, l'argument *valeur_proche* est complété obligatoirement par la valeur FAUX.

RECHERCHEH

- Dans « RECHERCHEH », la lettre H signifie « horizontale ».
- Cette fonction utilise généralement 2 tableaux : un 1^{er} tableau à compléter par le résultat de la formule ; un 2nd tableau contenant la valeur à renvoyer comme résultat. Ce second tableau est appelé « tableau de correspondance ».
- La fonction RECHERCHEH compte 4 arguments ; seul le 4^e est facultatif.
- L'argument *valeur_cherchée* désigne la valeur à retrouver dans la 1^{re} ligne du tableau de correspondance.
- L'argument *tableau* désigne le tableau de correspondance.
- L'argument *no_index_lig* utilise toujours un nombre entier positif désignant le rang de la ligne du tableau de correspondance contenant la valeur à retourner.
- L'argument *valeur_proche* n'est pas complété (ou complété par VRAI, de manière facultative) lorsque la 1^{re} ligne du tableau de correspondance contient des valeurs représentant des seuils ou paliers.
- La 1^{re} ligne du tableau de correspondance ne comporte normalement pas de doublons.

RECHERCHEH(valeur_cherchée; tableau; no_index_lig; valeur_proche)

=RECHERCHEH("Trimestre 2";A1:C3;3)

Trimestre 1	Trimestre 2	Trimestre 3
12	15	17
13	16	21

La 1^{re} ligne du tableau de correspondance placé en A1:C3 contient des valeurs textuelles classées dans l'ordre alphabétique (croissant).

La fonction RECHERCHEH cherche où se situe la valeur *Trimestre 2* dans la 1^{re} ligne du tableau de correspondance et renvoie la valeur située dans la 3^e ligne de cette colonne.

L'argument *valeur_proche* a été omis : il s'agit donc d'une recherche de valeur approximative rendue possible seulement car les trimestres sont classés dans l'ordre alphabétique.

Toutefois, l'argument *valeur_proche* aurait pu être complété avec la valeur logique FAUX afin d'effectuer une recherche de correspondance exacte entre la valeur cherchée et une valeur située dans la 1^{re} ligne du tableau.

Pour en savoir plus sur la manière de compléter l'argument *valeur_proche*, consulter la description de la fonction RECHERCHEV*.

INDEX (forme matricielle)

- Cette fonction renvoie, dans sa forme matricielle, une valeur en fonction de sa position dans une liste ou matrice.
- L'argument *matrice* peut être une constante ou, plus généralement, une plage de cellules.
- Les arguments *no_lig* et *no_col* ne peuvent pas être omis simultanément.
- Si la plage désignée comme matrice comporte à la fois plusieurs lignes et plusieurs colonnes, les arguments *no_lig* et *no_col* sont tous les deux requis.

EQUIV

- Cette fonction permet de renvoyer la position relative d'un élément recherché parmi une liste de valeurs.
- La plage sélectionnée ou la matrice de recherche doit comporter une seule ligne ou colonne.
- L'argument *type* permet d'indiquer s'il s'agit d'une correspondance exacte (indiquer 0) ou approximative (indiquer 1 pour rechercher dans une liste triée dans l'ordre croissant et -1 pour rechercher dans une liste triée dans l'ordre décroissant).

INDEX(matrice; no_lig; [no_col])

=INDEX(A1:B3;2;1)

Pierre	Brest
Paul	Nantes
Jacques	Rennes

Cette formule renvoie « Paul » car il s'agit de l'élément situé dans la 2^e ligne et dans la 1^{re} colonne.

EQUIV(valeur_cherchée;tableau_recherche; [type])

=EQUIV("Paul";{"Pierre"."Paul"."Jacques"};0)

Cette formule renvoie 2 car l'élément « Paul » se trouve en 2^e position dans la liste fournie sous forme de constante matricielle. Étant donné que les valeurs ne sont pas triées, seule la recherche d'une correspondance exacte est possible (type = 0)

=EQUIV(25;A1:A3;1)

=EQUIV(25;A1:A3;0)

10
20
30

La 1^{re} formule ci-contre renvoie 2 car la valeur 25 atteint 20 mais pas 30 et le type de recherche s'effectue dans une liste triée par ordre croissant.

La 2^e formule (type = 0 soit une correspondance exacte) renvoie #N/A car 25 ne figure pas dans la liste.

=EQUIV(25;A1:A3;-1)

30
20
10

Cette formule renvoie 1 car 25 est inférieur à 30 mais n'est pas inférieur à la valeur suivante, c'est-à-dire 20.

L'argument *type* étant -1, la fonction exige que les valeurs soient triées dans l'ordre décroissant. Si tel n'était pas le cas, la formule renverrait #N/A.

INDIRECT

- Cette fonction permet de composer une référence et de renvoyer la valeur ainsi désignée.
- L'argument *réf_texte* peut comporter des constantes et des références éventuellement concaténées avec l'opérateur & ou avec la fonction CONCATENER*.
- L'argument *a1* indique le style références : si cet argument facultatif est omis ou complété avec VRAI, les références utilisées sont de type A1 ; s'il est complété avec FAUX, le style de références est L1C1 (la cellule L3C4 correspond, dans le style A1, à la cellule C3).

INDIRECT(réf_texte;[a1])

=INDIRECT("A1")

Cette formule équivaut à =A1 et renvoie la valeur située dans la cellule A1 de la feuille active.

=INDIRECT(B1&"!A1")

Si la cellule B1 contient le texte « Février » et que le classeur comporte une feuille de calcul nommée « Février », la formule renvoie la valeur de la cellule A1 de la feuille « Février » et équivaut à =Février!A1.

=INDIRECT("B1&"!L1C1";FAUX)

Même exemple que ci-dessus mais en employant le style de références L1C1.

Ce style de référence est employé lorsqu'on souhaite bâtir la référence à l'aide, par exemple, de la fonction EQUIV*.

=INDIRECT("L"&EQUIV("Février";A1:A12;0)&"C4";FAUX)

Cette fonction INDIRECT emploie le style de références L1C1 dans lequel on fournit successivement le numéro de ligne et le numéro de la colonne : l'argument *a1* est FAUX.

La fonction INDIRECT contient ici une fonction EQUIV* dans l'argument *réf_texte*.

La lettre L est concaténée* avec la recherche de la position de la valeur « Février » dans une liste de valeurs avec EQUIV. Si la liste correspond aux mois de l'année triés dans l'ordre chronologique, la fonction renvoie 1 pour « Janvier », 2 pour « Février », 3 pour « Mars », etc.

Cette première partie de l'argument *réf_texte* est ensuite concaténée* avec le texte C4.

La fonction INDIRECT renvoie donc, dans cet exemple, la valeur de la cellule située dans la 2^e ligne et la 4^e colonne de la feuille active : L2C4 soit l'équivalent de D2.

AUJOURDHUI

- Cette fonction renvoie la date système, c'est-à-dire la date de l'appareil sur lequel est exécuté *Excel*.
- La fonction AUJOURDHUI n'a pas d'argument.

MAINTENANT

- Cette fonction renvoie la date-heure système, c'est-à-dire la date et l'heure de l'appareil sur lequel est exécuté *Excel*.
- Cette fonction n'a pas d'argument.

DATE

- Cette fonction renvoie une date à partir de 3 nombres : le 1^{er} pour l'année, le 2^e pour le mois et le 3^e pour le quantième du mois (jour).
- L'argument année est compris entre 1900 et 9999 ; l'argument mois est compris entre 1 et 12 ; l'argument jour entre 1 et 31.

AUJOURDHUI()

=AUJOURDHUI()	Renvoie la date du jour.
=AUJOURDHUI()-1	Renvoie la date de la veille.
=AUJOURDHUI()+7	Ajoute 7 jours à la date actuelle.

MAINTENANT()

=MAINTENANT()	Renvoie la date et l'heure au format jj/mm/aaaa hh :mm.
---------------	---

DATE(année; mois; jour)

=DATE(2017;3;31)	Cette formule renvoie 31/03/2017.
=DATE(2017;6;31)	Cette formule renvoie 01/07/2017 car le mois de juin compte seulement 30 jours.

ANNEE

Cette fonction renvoie un nombre compris entre 1900 et 9999 représentant l'année correspondant à la valeur de type date indiquée comme argument.

MOIS

Cette fonction renvoie un nombre compris entre 1 et 12 représentant le numéro du mois correspondant à la valeur de type date indiquée comme argument.

JOUR

Cette fonction renvoie un nombre compris entre 1 et 31 représentant le numéro du mois correspondant à la valeur de type date indiquée comme argument.

FIN.MOIS

- Cette fonction renvoie, à partir d'une valeur représentant une date (date au format jj/mm/aaaa, numéro de série, référence à une cellule contenant une date), la date du dernier jour du mois spécifié.
- L'argument *mois* indique le nombre de mois de décalage dans le futur (nombre positif) ou dans le passé (nombre négatif).

ANNEE(numéro_de_série)

=ANNEE("21/06/2017") Cette formule renvoie 2017.

=ANNEE(AUJOURDHUI()) Cette formule renvoie l'année de la date en cours.

MOIS(numéro_de_série)

=MOIS("21/06/2017") Cette formule renvoie 6.

=MOIS(AUJOURDHUI()) Renvoie le numéro correspondant au mois de la date en cours.

JOUR(numéro_de_série)

=JOUR("21/06/2017") Cette formule renvoie 21.

FIN.MOIS(date_départ; mois)

=FIN.MOIS("01/02/2017";0) Cette formule renvoie 28/02/2017.

=FIN.MOIS("01/02/2017";1) Cette formule renvoie 31/03/2017.

=FIN.MOIS("01/02/2017";-1) Cette formule renvoie 31/01/2017.

=FIN.MOIS(AUJOURDHUI();0) Cette formule renvoie la date du dernier jour du mois en cours.

NB.JOURS.OUVRES

- Cette fonction évalue le nombre de jours ouvrés (du lundi au vendredi) compris entre 2 dates en soustrayant, le cas échéant, des jours fériés non travaillés.
- Pour désigner des jours ouvrés personnalisés, il faut utiliser NB.JOURS.OUVRES.INTL.
- Les dates de départ et de fin peuvent être des numéros de série, obtenues avec la fonction DATE, des références à des cellules ou encore des constantes désignant des dates valides saisies au format jj/mm/aaaa et placées entre guillemets.

SERIE.JOUR.OUVRE

- Cette fonction renvoie le numéro série correspondant à une date située un nombre spécifié de jours ouvrés avant ou après une date donnée.
- La date de départ peut être un numéro de série, obtenue avec la fonction DATE*, une référence ou encore une constante désignant une date valide saisie au format jj/mm/aaaa et placée entre guillemets.
- Pour désigner des jours ouvrés personnalisés, il faut utiliser SERIE.JOUR.OUVRE.INTL.

NB.JOURS.OUVRES(date_départ; date_fin; [jours_fériés])

=NB.JOURS.OUVRES("01/01/2017";"31/01/2017")

Nombres de jours ouvrés (du lundi au vendredi) du mois de janvier 2017 sans prise en compte de jours fériés (le 1^{er} janvier de l'année considérée a eu lieu un dimanche).

=NB.JOURS.OUVRES("01/11/2017";"30/11/2017";{"01/11/2017"."11/11/2017"})

Nombre de jours ouvrés (du lundi au vendredi) du mois de novembre 2017 avec les 2 jours fériés nationaux de ce mois indiqués sous forme de constante matricielle (entre accolades).

=NB.JOURS.OUVRES("01/05/2017";"31/05/2017";D1:D3)

Nombre de jours ouvrés (du lundi au vendredi) du mois de mai 2017 avec l'argument *jours_fériés* complété par une référence à la plage D1:D3.

SERIE.JOUR.OUVRE(date_départ; nb_jours; [jours_fériés])

=SERIE.JOUR.OUVRE("10/07/2017";8)

Renvoie la date correspondant au 8^e jour ouvré après le 10 juillet 2017 sans prise en compte du 14 juillet : le résultat est 20/07/2017.

=SERIE.JOUR.OUVRE("10/07/2017";8;"14/07/2017")

Renvoie la date correspondant au 8^e jour ouvré après le 10 juillet 2017 en prenant en compte le vendredi 14 juillet : le résultat est 21/07/2017.

NO.SEMAINES

- Cette fonction renvoie le numéro de la semaine correspondant à une date.
- L'argument facultatif *type_renvoi* permet d'indiquer le mode de calcul : premier jour de la semaine, première semaine de l'année. Pour obtenir un calcul respectant la norme ISO 8601¹ en vigueur en Europe, utiliser 21.
- Microsoft met en garde contre des erreurs survenant lorsque la date est donnée sous forme de texte.

NO.SEMAINES.ISO

- Cette fonction renvoie le numéro de la semaine correspondant à une date en respectant la norme ISO 8601¹ observée en Europe.
- Cette fonction est utilisable depuis la version *Excel 2013*.

1. Norme précisant que la 1^{re} semaine d'une année civile est celle comportant au moins 4 jours dans la nouvelle année. Cette norme est également appelée « système de numérotation des semaines européen ».

NO.SEMAINES(date_série; [type_renvoi])

=NO.SEMAINES("01/01/2017")

Cette formule renvoie le numéro de semaine correspondant au dimanche 1^{er} janvier 2017, c'est-à-dire 1.

L'argument facultatif *type_renvoi* n'étant pas complété, le système utilisé considère comme 1^{re} semaine de l'année celle qui comprend le 1^{er} janvier avec une semaine commençant le dimanche.

Ce système ne correspond pas à la norme ISO 8601.

=NO.SEMAINES("01/01/2017";21)

Cette formule renvoie le numéro de semaine correspondant au dimanche 1^{er} janvier 2017, c'est-à-dire 52.

L'argument facultatif *type_renvoi* étant par complété par 21 (système 2) pour correspondre à la norme ISO 8601.

NO.SEMAINES.ISO(date)

=NO.SEMAINES.ISO("01/01/2017")

Cette formule renvoie le numéro de semaine correspondant au dimanche 1^{er} janvier 2017 en respectant la norme ISO 8601, c'est-à-dire 52.

MOIS.DECALER

Cette fonction renvoie une date (plus exactement son numéro de série) située un nombre spécifié de mois avant ou après une date donnée.

JOURSEM

- Cette fonction renvoie un nombre entier indiquant le rang du jour de la semaine correspondant à une date donnée.
- Cette fonction propose plusieurs modes de calcul, en particulier pour indiquer quel est 1^{er} jour de la semaine. Pour considérer le lundi comme 1 et le dimanche comme 7, indiquer 2.

MOIS.DECALER(date_départ; mois)

=MOIS.DECALER("21/06/2017";2)	Renvoie le numéro de série correspondant au 21 août 2017.
=MOIS.DECALER("21/06/2017";-3)	Renvoie le numéro de série correspondant au 21 mars 2017.
=MOIS.DECALER("30/06/2017";1)	Renvoie le numéro de série correspondant au 30 juillet 2017.
=MOIS.DECALER("31/05/2017";1)	Renvoie le numéro de série correspondant au 30 juin 2017.

JOURSEM(numéro_de_série; [type_retour])

=JOURSEM("21/06/2017") =JOURSEM("21/06/2017";1)	Ces fonctions renvoient 4 car le 21 juin 2017 se produit un mercredi qui est le 4 ^e jour de la semaine avec le mode de calcul considéré (dimanche premier jour de la semaine) avec l'argument <i>type_retour</i> = 1 ou omis.
=JOURSEM("21/06/2017";2)	Cette fonction renvoie 3 car le 21 juin 2017 se produit un mercredi qui est le 3 ^e jour de la semaine avec le mode de calcul considéré (l'argument <i>type_retour</i> est complété par la valeur 2 ce qui signifie que le lundi est considéré comme 1 ^{er} jour de la semaine).

DATEDIF

- Cette fonction effectue divers calculs d'intervalles entre 2 dates.
- C'est une fonction cachée d'Excel qui n'est donc pas documentée dans le logiciel (elle ne figure pas dans les bibliothèques de fonctions* de l'onglet *Formules*).
- C'est une fonction proposée par Excel en vue d'offrir une compatibilité avec le tableur Lotus 1-2-3.
- L'argument *date_début* doit être inférieur à l'argument *date_fin*.
- Les arguments *date_début* et *date_fin* sont des références à des cellules, des dates saisies sous formes de constantes du type "jj/mm/aaaa" ou obtenues à l'aide de la fonction DATE*.
- L'argument *type* sert à indiquer le type de calcul à effectuer (cf. tableau ci-contre).

DATEDIF(date_début; date_fin; type)

=DATEDIF("01/01/2017";"15/03/2017";"m")	Renvoie le nombre de mois entiers écoulés entre le 1 ^{er} janvier 2017 et le 15 mars 2017 soit 2 mois.
=DATEDIF("01/01/2017";"31/03/2017";"m")	Renvoie le nombre de mois entiers écoulés entre le 1 ^{er} janvier 2017 et le 31 mars 2017 soit 2 mois car le 31 mars est exclu du calcul.
=DATEDIF("23/09/1975";"01/06/2017";"y")	Calcule la différence en années complètes, c'est-à-dire l'âge ou l'ancienneté en années.
=DATEDIF("14/03/2010";"31/05/2017";"y")&" ans "&DATEDIF("14/03/2010";"31/05/2017";"ym")&" mois "&DATEDIF("14/03/2010";"31/05/2017";"md")&" jours"	Renvoie « 7 ans 2 mois 17 jours » (cf. ci-dessous)

Argument type	Type de calcul effectué
---------------	-------------------------

"y"	différence en années (<i>year</i>)
"m"	différence en mois (<i>month</i>)
"d"	différence en jours (<i>day</i>)
"ym"	différence en mois, après soustraction des années
"yd"	différence en jours, après soustraction des années
"md"	différence en jours, après soustraction des années et des mois

MAJUSCULE

Cette fonction convertit une chaîne textuelle en capitales d'imprimerie.

MINUSCULE

Cette fonction convertit une chaîne textuelle en minuscules.

NOMPROPRE

Cette fonction convertit la première lettre de chaque mot d'une chaîne textuelle en capitales d'imprimerie.

GAUCHE

- Cette fonction renvoie un nombre déterminé de caractères situés au début d'un texte.
- Lorsque l'argument facultatif *no_car* est omis, seul le 1^{er} caractère est renvoyé.

DROITE

- Cette fonction renvoie un nombre déterminé de caractères situés à la fin d'un texte.
- Lorsque l'argument facultatif *no_car* est omis, seul le dernier caractère est renvoyé.

MAJUSCULE(texte)

=MAJUSCULE("lundi") Renvoie « LUNDI ».

MINUSCULE(texte)

=MINUSCULE("LUNDI") Renvoie « lundi ».

NOMPROPRE(texte)

=NOMPROPRE("lundi 19 juin") Renvoie « Lundi 19 Juin ».

GAUCHE(texte; no_car)

=GAUCHE("Lundi") Renvoie « L ».
 =GAUCHE("Lundi";2) Renvoie « Lu ».
 =GAUCHE("Lundi";3) Renvoie « Lun ».

DROITE(texte; no_car)

=DROITE("Trimestre 1") Renvoie « 1 ».
 =DROITE("j.dupont@mpr.fr";2) Renvoie « fr ».
 =DROITE("FR-0023-420";3) Renvoie « 420 ».

STXT

- Cette fonction renvoie un nombre spécifié de caractères appartenant à une chaîne textuelle à partir du caractère occupant la position spécifiée.
- Si le nombre de caractères indiqué dans l'argument *no_départ* dépasse le nombre de caractères pouvant être extraits, la fonction renvoie une chaîne de texte vide.
- Si le nombre de caractères indiqué dans l'argument *no_car* dépasse le nombre de caractères présents dans le texte, tout le texte pouvant être extrait est renvoyé.

TEXTE

- Cette fonction convertit un nombre (dont une date ou une heure) en texte dans le format spécifié.
- L'argument *format_texte* utilise des codes de formats de nombres tels que ceux présentés dans l'onglet *Nombre* de la boîte de dialogue *Format de cellule*.

CNUM

Cette fonction convertit une valeur quelconque compatible en une valeur numérique.

STXT(texte; no_départ; no_car)

=STXT("F-35400 ST MALO";3;5) Renvoie 5 caractères à partir du 3^e caractère soit 35400

TEXTE(valeur; format_texte)

=TEXTE("01/08/2017";"jjjj") Convertit la date en texte en affichant le jour de la semaine en toutes lettres.

=TEXTE("1302,2387";"# ##0,00") Renvoie 1 302,24 (Séparateur de milliers et 2 décimales).

CNUM(texte)

=CNUM("01/01/2017") Renvoie 42736 (numéro de série correspondant à la date).

=CNUM ("1 500 €") Renvoie 1500.

=CNUM("12 kg") Renvoie #VALEUR! car la valeur ne peut être convertie.

CHERCHE

- Cette fonction effectue une recherche dans une chaîne de caractères et indique la position de début du texte recherché.
- Cette fonction n'est pas sensible à la casse.
- Si le texte à trouver n'existe pas, elle renvoie la valeur d'erreur #VALEUR!
- Cette fonction admet l'utilisation des caractères génériques * et ?.
- L'argument facultatif *no_départ* permet d'ignorer une partie de la chaîne dans laquelle est effectuée la recherche.

TROUVE

- Cette fonction effectue une recherche dans une chaîne de caractères et indique la position de début du texte recherché.
- Elle se distingue de la fonction CHERCHE par les points suivants :
 - elle est sensible à la casse ;
 - elle n'admet pas l'utilisation des caractères génériques * et ?.

CHERCHE(texte_cherché; texte; [no_départ])

=CHERCHE("-";"Cesson-Sévigné")	Renvoie 7 car un trait d'union occupe la 7 ^e position dans la chaîne de caractères « Cesson-Sévigné ».
=CHERCHE(CAR(160);A2)	La fonction CAR() permet de désigner un caractère par son code numérique. Cette fonction renverra un nombre s'il existe au moins une espace insécable dans A2 (code 160).
=CHERCHE("S";"Rennes")	Renvoie 6 car il s'agit de la 6 ^e lettre du mot « Rennes », sans tenir compte de la casse.
=ESTNUM(CHERCHE("a";A2))	Cette formule renvoie VRAI si la lettre <i>a</i> est présente dans la cellule A2 : la fonction ESTNUM* évalue si le résultat de la fonction CHERCHE est un numéro de position ou bien une valeur d'erreur (lorsque la lettre <i>a</i> ne figure pas dans A2).
=CHERCHE("(*)";A1)	Cette formule utilise le caractère générique * afin d'évaluer s'il existe du texte entre parenthèses dans la cellule A1 et renvoyer la position de la parenthèse ouvrante.

TROUVE(texte_cherché; texte; [no_départ])

=TROUVE("S";"Rennes")	Renvoie #VALEUR! car la lettre recherchée est un <i>s</i> majuscule et que le texte dans lequel est effectuée la recherche ne contient qu'un <i>s</i> minuscule.
-----------------------	--

SUBSTITUE

- Cette fonction remplace un ou plusieurs caractères dans une chaîne de texte par aucun, un ou plusieurs caractères spécifiés.
- L'argument *texte* désigne la chaîne de caractères dans laquelle effectuer le remplacement (référence à une cellule, constante textuelle).
- *Ancien_texte* désigne le caractère ou la chaîne de caractères à remplacer.
- *Nouveau_texte* désigne le caractère ou la chaîne de caractères de remplacement.
- L'argument facultatif *no_position* indique l'occurrence de l'ancien texte qui doit être remplacée. Lorsqu'il n'est pas complété, toutes les occurrences de l'ancien texte sont remplacées.
- Cette fonction est sensible à la casse.

NBCAR

Cette fonction renvoie le nombre de caractères contenus dans une chaîne de caractères de tous types, espaces inclus.

SUBSTITUE(texte; ancien_texte; nouveau_texte; [no_position])

=SUBSTITUE(A2;" ";"")	Renvoie la chaîne de caractères contenue dans A2 débarrassée des espaces.
=SUBSTITUE(A2;CAR(160);"")	Renvoie la chaîne de caractères contenue dans A2 débarrassée des espaces insécables
=SUBSTITUE(A2;0;" +33";1)	Renvoie la chaîne de caractères contenue dans A2 sans le zéro initial (l'argument <i>no_position</i> est complété par 1 pour indiquer qu'on ne veut pas supprimer tous les zéros mais seulement le premier) et remplace le zéro par « +33 ». Ce traitement est applicable à des numéros de téléphone pour les présenter au format international.
=SUBSTITUE(A2;"A";"B") =SUBSTITUE(A2;"a";"b")	Ces 2 formules ne sont pas équivalentes car SUBSTITUE est une fonction sensible à la casse.
=SUBSTITUE("Rennes";"z";"a")	Cette fonction renvoie la chaîne initiale car aucune substitution n'est possible.

NBCAR(texte)

=NBCAR("Rennes")	Renvoie 6 car le mot comporte 6 lettres
=NBCAR("35000")	Renvoie 5 car la valeur numérique comporte 5 chiffres.
=NBCAR("35000 Rennes")	Renvoie 12 car il y a 5 chiffres pour le code postal, un espace et 6 lettres pour le mot.

CONCATENER

- « Concaténer » signifie « enchaîner » lorsqu'il s'agit de caractères.
- Cette fonction assemble des éléments disparates (textes, nombres, références, etc.).
- Elle comporte 1 à 255 arguments.

SUPPRESPEACE

- Cette fonction supprime tous les espaces en surnombre dans une chaîne de caractères pour ne conserver que des espaces uniques entre les mots.
- Cette fonction ne supprime pas les espaces insécables.

EXACT

- Cette fonction compare 2 textes en respectant la casse et renvoie VRAI lorsqu'ils sont identiques et FAUX dans le cas contraire.

CONCATENER(texte1; [texte2]; ...)

=CONCATENER(A2;B2;C2)	Équivaut à A2&B2&C2
=CONCATENER(A2;" ";B2)	Assemble la valeur de A2, un espace et la valeur de B2. Équivaut à A2&" "&C2

SUPPRESPEACE(texte)

=SUPPRESPEACE(A2)	Cette formule élimine tous les espaces inutiles au début, à la fin et entre les mots contenus dans A2.
=SUPPRESPEACE("35000 RENNES")	Renvoie « 35000 RENNES » (il y a des espaces en surnombre entre le code postal et le nom de la ville).

EXACT(texte1; texte2)

=EXACT("mardi";"Mardi")	Renvoie FAUX.
=EXACT("mardi";"mardi")	Renvoie VRAI.
=EXACT(A2;"Mardi")	Cette formule compare la valeur de A2 avec « Mardi ».


GÉNÉRALITÉS	3
FONCTIONS MATHÉMATIQUES	4
SOMME	4
PRODUIT	4
ARRONDI	5
ENT	5
ABS	5
SOMME.SI	6
SOMME.SI.ENS	7
SOMMEPROD	8
SOUS.TOTAL	9
FONCTIONS STATISTIQUES	10
MOYENNE	10
MAX	10
MIN	10
NB	11
NBVAL	11
NB.VIDE	11
NB.SI	12
NB.SI.ENS	13
EQUATION.RANG	14
GRANDE.VALEUR	15
PETITE.VALEUR	15
FONCTIONS LOGIQUES	16
SI	16
SIERREUR	16
SI.NON.DISP	17
NON	17


ET	18
OU	18
FONCTIONS D'INFORMATION	19
ESTVIDE	19
ESTNA	19
ESTNUM	19
FONCTIONS DE RECHERCHE ET RÉFÉRENCE	20
RECHERCHEV	20
RECHERCHEH	21
INDEX (forme matricielle)	22
EQUIV	22
INDIRECT	23
FONCTIONS DE DATE ET HEURE	24
AUJOURDHUI	24
MAINTENANT	24
DATE	24
ANNEE	25
MOIS	25
JOUR	25
FIN.MOIS	25
NB.JOURS.OUVRES	26
SERIE.JOUR.OUVRE	26
NO.SEMAIN	27
NO.SEMAIN.ISO	27
MOIS.DECALER	28
JOURSEM	28
DATEDIF	29

FONCTIONS DE TEXTE.....	30
MAJUSCULE	30
MINUSCULE	30
NOMPROPRE.....	30
GAUCHE.....	30
DROITE.....	30
STXT	31
TEXTE.....	31
CNUM.....	31
CHERCHE.....	32
TROUVE	32
SUBSTITUE.....	33
NBCAR.....	33
CONCATENER	34
SUPPRESPECE	34
EXACT	34
COMMENT UTILISER CE FICHER ?.....	37




← Comment utiliser ce fichier ?

Depuis le sommaire

- Cliquez sur  pour accéder à la table des matières détaillée et interactive.
- Cliquez sur les tuiles pour accéder au début de chaque chapitre.



À tout moment

- Cliquez sur  pour revenir au sommaire.
- Cliquez sur les renvois  pour obtenir plus de détails sur les points évoqués ;
- utilisez **Alt** +  pour revenir à la page précédemment affichée.



Pour rechercher

- Dans le menu **Édition**, cliquez sur **Rechercher** ou réalisez la combinaison **Ctrl** + **F**.
- Tapez le(s) terme(s) recherché(s) dans la zone de recherche et validez par **Entrée** autant de fois que nécessaire pour atteindre le résultat attendu.

